
Subject: Re: Anyway to avoid this last for loop
Posted by [Tom McGlynn](#) on Thu, 08 Jan 2009 14:28:02 GMT
[View Forum Message](#) <> [Reply to Message](#)

On Jan 8, 4:59 am, hlde...@gmx.de wrote:

```
> ;AClength is how many frames are thrown out after one AC Hit ->
> multiple entries can fall into one bin
>
> detHist=histogram(data,REVERSE_INDICES=ri, /L64, binSize=AClength)
> acHist=histogram(ACdeletes, /L64, binSize=AClength)
>
> ;renormalize to one
> dl=where(detHist GT 1, cntD)
> al=where(acHist GT 1, cntA)
> IF cntD NE 0 THEN detHist[dl]=1
> IF cntA NE 0 THEN acHist[al]=1
```

Can't these last four lines be replaced by

```
detHist = detHist<1
acHist = acHist<1
```

The < and > operators are useful though not necessarily intuitive.

```
>
> ;subtract the two histograms. All detector frames which have
> corresponding AC frame should now have value 0, if AC frame exists but
> not detector frame then value is -1. If only detector frame exists
> (the ones we want) value stays 1!!!
> diffHist=detHist-acHist
>
> ;keep the ones with one
> keep=where(diffHist EQ 1, cnt)
>
> ;And now my problem. I need the indices of the keep-frames but there
> can be more than one in a bin so I'm stuck with this loop:
>
> FOR k=0L, cnt-1 DO BEGIN
>     keepIndices=[keepIndices,ri[ri[keep[k]]:ri[keep[k]+1]-1]]
> ENDFOR
>
```

To do it all at once, I think you can just go back to the data array as below:

```
keepHist = where (diffHist EQ 1, cnt) ; From the original
if (cnt gt 0) then begin
    keepers = intarr(n_elements(detHist))
    keepers[keepHist] = 1;
```

```
    keepIndices = where(keepers[data/ACLENGTH] eq 1)
endif else begin
    keepIndices = [-1]
endelse
```

Haven't tested it, but it seems like it or something like it should work reasonably efficiently.

Good luck,
Tom McGlynn
